
Inleiding

Het vertalen van de gebruikersinterface in code met behulp van ms access 

Iedereen heeft wel eens dat dilemma; wordt de gebruikersinterface in het engels ontwikkeld of wordt er voor een andere taal gekozen. Visual Basic is in het engels; ook daarom zal de gebruikersinterface meestal in het engels geschreven zijn. Om de interface snel en makkelijk te kunnen vertalen maak ik gebruik van de volgende werkwijze. In een database worden twee tabellen toegevoegd: Language met als velden LanguageId (type Number) en LanguageText (type Text). Deze twee velden vormen ook de sleutel. De tweede tabel heeft als naam Translate en heeft in beginsel slechts twee velden: Source (type Text) en 1 (type Text). In het veld Source staat de unieke omschrijving van het te vertalen object. De veldnaam 1 verwijst naar de LanguageId. In dit voorbeeld is de tabel Language gevuld met een record namelijk: LanguageId = 1; LanguageText = English.

In de code moet er een procedure komen die het mogelijk maakt om een taal aan de applicatie toe te voegen en daarbij tegelijkertijd in de tabel Translate een nieuw record aan te maken.

Private Function InsertLanguage(ByRef colTMP As  

 Collection) As Long

 Dim strValue As String

 Dim lngLanguageId As Long

 On Error GoTo InsertLanguage_FAIL

 '

 strValue = colTMP("LanguageText")

 lngLanguageId = GetHighestId("LanguageId", _ 

"Language")

 strSQL = "SELECT LanguageId, LanguageText " + _

   " FROM Language WHERE LanguageText = '" + _  

   strValue + "'"

 Set rs = db.OpenRecordset(strSQL)

 With rs

  If Not (.BOF And .EOF) Then

   .Edit

   .Fields("LanguageText").Value = strValue

   .Update

  Else

   .AddNew

   .Fields("LanguageId").Value = lngLanguageId

   .Fields("LanguageText").Value = strValue

   .Update

   Call AppendLanguage(lngLanguageId)

  End If

 End With

 rs.Close

 InsertLanguage = 0

 Exit Function

InsertLanguage_FAIL:

 InsertLanguage = -1

End Function

Private Function AppendLanguage(ByRef lngValue As 

 Long)

 Dim TempField As New Field

 '

 With TempField

  .Name = CStr(lngValue)

  .Type = dbText

  .Size = 255

 End With

 db.TableDefs("Translate").Fields.Append TempField

End Function

Hierna is het vrij simpel om alle gebruikte controls op een form op te halen. De control-naam in combinatie met de form-naam vormt dan de sleutel (het veld Source in de tabel Translate).

Public Sub CollectStrings(ByRef xForm As Form)

'THIS procedure collects all captions and text in 'forms and module and put them into the db ONLY call 'once! and perhaps after major changes in code

 Dim rs As Recordset

 Dim xCtrl As Control

 Dim strSQL As String

 '

 For Each xCtrl In xForm.Controls

  If TypeOf xCtrl Is Label Or _

   TypeOf xCtrl Is Form Or _

   TypeOf xCtrl Is Menu Or _

   TypeOf xCtrl Is CheckBox Or _

   TypeOf xCtrl Is OptionButton Or _

   TypeOf xCtrl Is CommandButton Then

   '

   strSQL = "SELECT * FROM Translate " + _

     " WHERE Source = '" & xForm.Name + "." + _

     xCtrl.Name & "' "

   Set rs = db.OpenRecordset(strSQL)

   With rs

    If (.BOF And .EOF) Then

     .AddNew

     .Fields("Source").Value = xForm.Name + _

       "." + xCtrl.Name

     .Fields("1").Value = xCtrl.Caption

     .Update

    End If

   End With

   rs.Close

  End If

 Next

End Sub

Deze procedure hoeft eigenlijk maar een (1) keer te worden uitgevoerd. De beste plaats lijkt mij het LOAD event en dan nog voor het daadwerkelijke aanmaken en uitleveren van een executable. Alleen als er wijzigingen zijn in het form voor wat betreft de gebruikte controls moet deze procedure opnieuw worden aangeroepen.

De tabel Translate is nu gevuld met alle voorkomende controls op een form met als waarde in veld 1 de engelse tekst (Caption). Om nu een vertaling te kunnen krijgen hebben we - of met behulp van de applicatie of handmatig direct in ms access - een tweede taal toegevoegd. De tabel Translate bestaat dan uit drie velden: Source de unieke sleutel; 1 wat de Engelstalige omschrijving bevat en veld 2. Hierin komt de vertaling. Het vertalen zelf kan weer of handmatig in ms access worden uitgevoerd of in een bepaalde functie van de applicatie (zie het hiervoor het uitgewerkte voorbeeld in de zip).

Bij het opstarten van het betreffende form moeten alle beschikbare en relevante controls worden vertaald. Door het zoeken naar de sleutel in de database wordt de relevante vertaling opgehaald.

Public Sub TranslateForm(ByRef xForm As Form)

 Dim rs As Recordset

 Dim xCtrl As Control

 Dim intX As Integer

 Dim strSQL As String

 '

 For Each xCtrl In xForm.Controls

  If TypeOf xCtrl Is Label Or _

   TypeOf xCtrl Is Form Or _

   TypeOf xCtrl Is Menu Or _

   TypeOf xCtrl Is CheckBox Or _

   TypeOf xCtrl Is OptionButton Or _

   TypeOf xCtrl Is CommandButton Then

   '

   strSQL = "SELECT * FROM Translate " + _

    " WHERE Source = '" & xForm.Name & _ 

    "." & xCtrl.Name & "'"

   Set rs = db.OpenRecordset(strSQL)

   With rs

    If Not (.BOF And .EOF) Then

     If IsNull(.Fields(lngLanguage).Value) Or _

Len(.Fields(lngLanguage).Value) = 0 Then

xCtrl.Caption = .Fields("1").Value & strEmpty

     Else

       xCtrl.Caption = .Fields(lngLanguage).Value

    End If

   Else

    .AddNew

    .Fields("Source").Value = xForm.Name & _

     "." & xCtrl.Name

    .Fields("1").Value = xCtrl.Caption

    .Update

   End If

  End With

 End If

 Next

End Sub

De variabele 'lngLanguage' is hier de gekozen taal (LanguageId). Is de omschrijving van het betreffende control niet vertaald of afwezig dan wordt automatisch de Engelstalige (de oorspronkelijke) omschrijving gebruikt. Op deze wijze worden snel de controls op een form van een omschrijving in de juiste taal voorzien.

Er zijn echter ook andere teksten die gebruikt worden en niet tot een control te herleiden zijn. Teksten in een messagebox bijvoorbeeld; maar ook teksten die in een module worden gezet.

Hiervoor is er geen unieke sleutel beschikbaar. Er zit dan niet anders op om dit zelf aan te geven. Daarvoor wordt gebruik gemaakt van een aparte schrijfwijze in code:

Call MsgBox(TRANSLATESTRING("Phone not inserted", _ "SavePhone"), vbOKOnly, App.FileDescription)

TRANSLATESTRING is een functie die zoekt naar een sleutel (SavePhone) in de tabel Translate. Bestaat deze niet dan wordt deze toegevoegd met de omschrijving (Phone not inserted).

Public Function TRANSLATESTRING(ByRef strValue As String, ByRef strKey As String) As String

 Dim strSQL As String

 Dim rs As Recordset

 '

 strSQL = "SELECT * FROM Translate " + _

  " WHERE Source = '" & strKey & "'"

 Set rs = db.OpenRecordset(strSQL)

 With rs

  If Not (.BOF And .EOF) Then

   If IsNull(.Fields(lngLanguage).Value) Then

    TRANSLATESTRING = .Fields("1").Value & strEmpty

   Else

    TRANSLATESTRING = .Fields(lngLanguage).Value

   End If

  Else

   .AddNew

   .Fields("Source").Value = strKey

   .Fields("1").Value = strValue

   .Update

  End If

 End With

End Function

Het is makkelijker om eerst alle sleutels uit de verschillende forms en modulen op te halen voordat er wordt vertaald. Een eenvoudig zoekmechanisme haalt alle betreffende teksten op en slaat deze onder de juiste sleutel op in de database. Ook deze procedure is slechts eenmalig nodig (behoudens wijzigingen in de code).

Public Sub ReadModuleFiles()

 Dim intFN As Integer

 Dim strFileName As String

 Dim strLine As String

 Dim pos1 As Integer

 Dim pos2 As Integer

 Dim intX As Integer

 Dim aModuleName(2) As String

 Dim strModuleName As String

 Const strTRANSLATESTRING As String =   

"TRANSLATESTRING"

 '

 Screen.MousePointer = vbHourglass

 aModuleName(0) = "modGlobal.bas"

 aModuleName(1) = "modSQL.bas"

 aModuleName(2) = "frmTranslate.frm"

 For intX = LBound(aModuleName) To UBound(aModuleName)

  strModuleName = aModuleName(intX)

  intFN = FreeFile

  strFileName = App.Path & "\" & strModuleName

  Open strFileName For Input As intFN

  Do While Not EOF(intFN)

   Line Input #intFN, strLine

   pos1 = InStr(UCase(strLine), strTRANSLATESTRING)

   If pos1 <> 0 Then

    pos2 = InStr(Right(strLine, Len(strLine) - pos1+ _

      1 - Len(strTRANSLATESTRING)), """)")

    Call StripString(Mid(strLine, pos1 + _

      Len(strTRANSLATESTRING), pos2 + 1))

   End If

  Loop

  Close #intFN

 Next

 Screen.MousePointer = vbDefault

End Sub

Private Function StripString(ByRef strSource As String) As String

    Dim intX As Integer

    Dim rs As Recordset

    Dim strKey As String

    Dim strValue As String

    Dim strTMP As String

    Dim strSQL As String

    '

    strTMP = ExtractString(strSource, 1)

    strValue = Mid(strTMP, 3, Len(strTMP) - 3)

    strTMP = ExtractString(strSource, 2)

    strKey = Mid(strTMP, 3, Len(strTMP) - 4)

    strSQL = "SELECT * FROM Translate " + _

        " WHERE Source = '" & strKey & "' "

    Set rs = db.OpenRecordset(strSQL)

    With rs

        If (.BOF And .EOF) Then

            .AddNew

            .Fields("Source").Value = strKey

            .Fields(1).Value = strValue

            .Update

        End If

    End With

    rs.Close

End Function

Op deze manier zijn op een simpele manier meertalige applicaties te maken. De tijd die gemoeid is met op ophalen van de vertaling uit de database is met de huidige kloksnelheden te verwaarlozen. Het is nu ook mogelijk om het aan de gebruikers zelf over te laten om de applicatie in de door hen gewenste taal te vertalen.

Bovenstaande procedures en functies zijn gekopieerd uit een werkend voorbeeld project dat te vinden op http://www.kather.net/VisualBasicSource/Publish















